

DBPedia feladatok

elérés: <http://dbpedia.org/sparql>

1 – Próbáljuk ki a DBPedia böngészőjét! Nézzük meg a következő URL-t:

http://dbpedia.org/page/Eötvös_Loránd_University

- a – Keressük meg a SameAs property-vel kapcsolódó oldalakat. Nézzük meg a hivatkozásokat.
- b – Keressük meg a képet tartalmazó property-t.
- c – Nézzük meg az egyetem dolgozóit.

2 – Kérdezzük le az első 1000 típust a dbpedia-ról!

3 – SPARQL: Keressünk olyan dolgozót akinek van youtube videója.

(FILTER (Contains(„ebben keresünk”, „ezt”)))

str(<dbpedia-owl:abstract>)

4 – Nézzük meg a Talált dolgozó adatait ntriples, xml és n3/turtle formátumba.

Tweet feladatok

MINDENHOL HASZNÁLJUNK LIMIT 10-ET A LEKÉRDEZÉS VÉGÉN!!!!

SWP portálon. (elérés: <http://eszakigrd107.inf.elte.hu:7001/SemWeb-war/>)

1 - Alap Twitter lekérdezés példa.

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
SELECT ?tweet ?user_id ?lang ?text ?lat ?lon ?url
WHERE {
    ?tweet a tw:Tweet .
    ?tweet tw:user_id ?user_id .
    ?tweet tw:lang ?lang .
    ?tweet tw:text ?text .
    OPTIONAL { ?tweet tw:url ?url } .
    OPTIONAL { ?tweet geo:lat ?lat } .
    OPTIONAL { ?tweet geo:long ?lon } .
} LIMIT 10
```

2 - Kérdezzük le a magyar tweetek azonosítóit és az üzenetüket!

3 – Kérdezzünk le country és lat, lon információkat a dbpedia-ről a példa query és a link alapján! Link:
<http://dbpedia.org/page/Hungary>.

Példa lekérdezés:

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?abs
WHERE {
    SERVICE <http://dbpedia.org/sparql> {
        <http://dbpedia.org/resource/Hungary> dbpedia-owl:abstract ?abs .
    } .
} LIMIT 10
```

4 – Kérdezzük le a lat: 13.81444309, lon: 100.65446956 koordinátaéhoz melyik country tartozik! Segítség:
közelítés +-0.5 koordináta

5 – Kérdezzük le melyik Tweet melyik országból érkezett!

6 - Az url-eket melyik országból tweetelték?

MEGOLDÁSOK

DBPEDIA

3 - Keressünk olyan dolgozót akinek van youtube videója

```
SELECT * WHERE {
  ?valaki <http://dbpedia.org/property/workplaces>
    <http://dbpedia.org/resource/E%C3%B6tv%C3%B6s_Lor%C3%A1nd_University> .
  ?valaki <http://dbpedia.org/ontology/wikiPageExternalLink> ?link .
  FILTER (contains(str(?link), "youtube"))
}
```

Twitter

1 - Alap Twitter lekérdezés példa.

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
SELECT ?tweet ?user_id ?lang ?text ?lat ?lon ?url
WHERE {
  ?tweet a tw:Tweet .
  ?tweet tw:user_id ?user_id .
  ?tweet tw:lang ?lang .
  ?tweet tw:text ?text .
  OPTIONAL { ?tweet tw:url ?url } .
  OPTIONAL { ?tweet geo:lat ?lat } .
  OPTIONAL { ?tweet geo:long ?lon } .
} LIMIT 10
```

Time: 2000 ms

2 - Kérdezzük le a magyar tweetek azonosítóit és az üzenetüket!

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
SELECT ?tweet ?text
WHERE {
  ?tweet a tw:Tweet .
  ?tweet tw:lang tw:hu .
  ?tweet tw:text ?text .
} LIMIT 10
```

Time: 147 ms

3 – Kérdezzünk le country és lat, lon információkat a dbpedia-ról a példa query és a link alapján!

Link: <http://dbpedia.org/page/Hungary>.

Példa lekérdezés:

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?abs
WHERE {
    SERVICE <http://dbpedia.org/sparql> {
        <http://dbpedia.org/resource/Hungary> dbpedia-owl:abstract ?abs .
    } .
} LIMIT 10
```

Megoldás:

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?country ?lat ?lon
WHERE {
    SERVICE <http://dbpedia.org/sparql> {
        ?country a dbpedia-owl:Country .
        ?country geo:lat ?lat .
        ?country geo:long ?lon .
    } .
} LIMIT 10
```

Time: 507 ms

4 – Kérdezzük le a lat: 13.81444309, lon: 100.65446956 koordinátahoz melyik country tartozik!

Segítség: közelítés +/-0.5 koordináta

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?country ?lat ?lon
WHERE {
    SERVICE <http://dbpedia.org/sparql> {
        ?country a dbpedia-owl:Country .
        ?country geo:lat ?lat .
        ?country geo:long ?lon .
        FILTER ( (?lat > (13.81444309-0.5)) && (?lat < (13.81444309+0.5))
            && (?lon > (100.65446956-0.5)) && (?lon < (100.65446956+0.5)) ) .
    } .
} LIMIT 10
```

Time: 90 ms

5 – Kérdezzük le melyik Tweet melyik országból érkezett!

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?tweet ?Tlat ?Tlon ?country
WHERE {
    ?tweet a tw:Tweet .
    ?tweet geo:lat ?Tlat .
    ?tweet geo:long ?Tlon .
    SERVICE <http://dbpedia.org/sparql> {
        ?country a dbpedia-owl:Country .
        ?country geo:lat ?lat .
        ?country geo:long ?lon .
    } .
    FILTER ( (?lat > (?Tlat-0.5)) && (?lat < (?Tlat+0.5))
            && (?lon > (?Tlon-0.5)) && (?lon < (?Tlon+0.5)) ) .
} LIMIT 10
Time: 3633 ms
```

6 - Az url-eket melyik országból tweetelték?

```
PREFIX tw: <http://triple.twitter.elte.hu/>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?tweet ?url ?country
WHERE {
    ?tweet a tw:Tweet .
    ?tweet geo:lat ?Tlat .
    ?tweet geo:long ?Tlon .
    ?tweet tw:url ?url
    SERVICE <http://dbpedia.org/sparql> {
        ?country a dbpedia-owl:Country .
        ?country geo:lat ?lat .
        ?country geo:long ?lon .
    } .
    FILTER ( (?lat > (?Tlat-0.5)) && (?lat < (?Tlat+0.5))
            && (?lon > (?Tlon-0.5)) && (?lon < (?Tlon+0.5)) ) .
} LIMIT 10
Time: 47145 ms
```